# Geometry of Complex Mappings

In this blog entry I like to show you how complex mappings can be visualized in EMT. For the first experiments, we are interested in the image f(D) of regions D of the complex plane. The function we use is plot2d() with a complex grid as a parameter.

The first grid is the rectangular region

$$r+i\phi, \qquad 0 \le \phi \le 2\pi, \quad 0 \le r \le 1$$

If we map this with the exponential function, we get a circular ring. That is the set of points

$$\rho \, (\cos(\phi )+i\sin(\phi)), \qquad 1 \le \rho \le e, \quad 0 \le \phi \le 2\pi.$$

We plot only 10 grid lines from the 100 available lines in the complex matrix containing the images of the grid points in the rectangular grid we started with.

>phi=0:0.01:1; r=(0:0.01:1)'; z=exp(r+2*pi*phi*I);
>plot2d(z,cgrid=10,r=3,grid=6):

We want to apply the Joukowsky mapping to this region. This mapping projects the unit circle to the interval [-1,1] with an orthogonal projection. The circular region becomes an ellipse with focal points -1 and 1 and sum of half axes e.

>plot2d((z+1/z)/2,cgrid=10,r=2,grid=6); plot2d([-1,1],0,thickness=4,>add):

The grid lines are all ellipses with focal points -1 and 1, or hyperbolas with the same focal points. The vertical and horizontal grid lines intersect orthogonally, because holomorphic functions are locally orthogonal.

Let us try another example. We take the simple square grid

$$R = [-1,1] \times [-1,1]$$

and map the corners to the points 0,1,2,3 with a polynomial of degree 3. This time, we do not take the lazy approach with a very fine grid. Instead, we plot the horizontal and vertical grid lines separately with different colors.

>z0=[1+I,-1+I,-1-I,1-I]; w0=complex(0:3);
>d=interp(z0,w0);
>x1=-1:0.1:1; x2=-1:0.01:1;
>z=x1'+I*x2; w=interpval(z0,d,z);
>plot2d(re(w),im(w),color=green,a=0,b=3,c=-1.5,d=1.5);
>z=x2'+I*x1; w=interpval(z0,d,z);
>plot2d(w0,>points,>add):

If you try to understand this more difficult plot you should realize the plot2d() is now used for two real matrices. corresponding rows of the first matrix and of the second matrix are plotted as a curve in the plane. We first take a 21 curves in horizontal, then 11 lines in vertical direction.

Clearly the square R is not mapped one-to-one. For such an example, we take transformation of a circle containing R to the upper half plane.

>x=-1:0.01:1; z=x+I*x'; z=z/sqrt(2);
>plot2d((z+I)/(1+I*z),cgrid=10,a=-4,b=4,c=0,d=8):

We often see another type of plot, where the real part and the complex part are color coded. In EMT, this can be done as follows.

We plot the real part and the imaginary part of the complex sine function in some region in the complex plane, including level lines for integer values of the function from -10 to 10.

>x=linspace(0,2pi,200); z=x+I*x';
>plot2d("re(sin(x+I*y))",a=0,b=2pi,c=-pi,d=pi, ...
>      levels=-10:10,>hue,>spectral,n=100,grid=6):

>plot2d("im(sin(x+I*y))",a=0,b=2pi,c=-pi,d=pi, ...
>     levels=-10:10,>hue,>spectral,n=100,grid=6):

Diese Website verwendet Akismet, um Spam zu reduzieren. Erfahre mehr darüber, wie deine Kommentardaten verarbeitet werden.